
Solutions

• Not all possible solutions are given.

• In many places, much less detail than I have provided in the example solution
was actually required.

• Solutions may contain typos.

Dit tentamen is in elektronische vorm beschikbaar gemaakt door de TBC van A–Eskwadraat.
A–Eskwadraat kan niet aansprakelijk worden gesteld voor de gevolgen van eventuele fouten
in dit tentamen.

Contracts (48 points total, plus 5 bonus points)

Here is a GADT of contracts:

data Contract :: ∗ → ∗where
Pred :: (a→ Bool)→ Contract a
Fun :: Contract a→ Contract b→ Contract (a→ b)

A contract can be a predicate for a value of arbitrary type. For functions, we offer
contracts that contain a precondition on the arguments, and a postcondition on the
results.

Contracts can be attached to values by means of assert. The idea is that assert will
cause run-time failure if a contract is violated, and otherwise return the original result:

assert :: Contract a→ a→ a
assert (Pred p) x = if p x then x else error "contract violation"
assert (Fun pre post) f = assert post ◦ f ◦ assert pre

For function contracts, we first check the precondition on the value, then apply the
original function, and finally check the postcondition on the result.

For example, the following contract states that a number is positive:

pos :: (Num a, Ord a)⇒ Contract a
pos = Pred (>0)

We have

assert pos 2 ≡ 2
assert pos 0 ≡ ⊥ (contract violation error)

1 (6 points). Define a contract

true :: Contract a

such that for all values x, the equation assert true x ≡ x holds. Prove this equation using
equational reasoning. •

Solution 1.

true = Pred (const True)

The proof:

assert true x
≡ { definition of true }

assert (Pred (const True)) x
≡ { definition of assert }

if (const True) x then x else error "contract violation"
≡ { definition of const }

if True then x else error "contract violation"
≡ { if True }

x

◦
Often, we want the postcondition of a function to be able to refer to the actual argu-

ment that has been passed to the function. Therefore, let us change the type of Fun:

Fun :: Contract a→ (a→ Contract b)→ Contract (a→ b)

The postcondition now depends on the function argument.

2 (4 points). Adapt the function assert to the new type of Fun. •
Solution 2.

assert (Fun pre post) f = λx→ (assert (post x) ◦ f ◦ assert pre) x

Haskell actually forces all clauses of a function definition to have the same number of
arguments, so to be entirely correct we have to use a lambda abstraction. But putting
the x to the left of = counts as correct, too. ◦
3 (4 points). Define a combinator

(_) :: Contract a→ Contract b→ Contract (a→ b)

that reexpresses the behaviour of the old Fun constructor in terms of the new and more
general one. •
Solution 3.

(_) pre post = Fun pre (const post)

◦
4 (6 points). Define a contract suitable for the list index function (!!), i.e., a contract of
type

Contract ([a ]→ Int→ a)

that checks if the integer is a valid index for the given list. •
Solution 4. We need a nested Fun application to get a function contract taking two
arguments. The actual condition is on the integer, so both the precondition for the first
argument and the postcondition are true.

lookupContract :: Contract ([a ]→ Int→ a)
lookupContract = Fun true (λxs→

Fun (Pred (λn→ 0 6 n ∧ n < length xs)) (λn→
true))

◦

5 (6 points). Define a contract

preserves :: Eq b⇒ (a→ b)→ Contract (a→ a)

where assert (preserves p) f x fails if and only if the value of p x is different from the
value of p (f x). Examples:

assert (preserves length) reverse "Hello" ≡ "olleH"
assert (preserves length) (take 5) "Hello" ≡ "Hello"
assert (preserves length) (take 5) "Hello world" ≡ ⊥

•
Solution 5.

preserves f = Fun true (λx→ Pred (λr→ f x = = f r))

◦

6 (6 points). Consider

preservesPos = preserves (>0)
preservesPos′ = pos _ pos

Is there a difference between assert preservesPos and assert preservesPos′? If yes, give an
example where they show different behaviour. If not, try to prove their equality using
equational reasoning. •

Solution 6. Both contracts have the same type, but they behave differently. Here is an
example:

example c = assert c id 0

With this definition, we get

example preservesPos ≡ 0
example preservesPos′ ≡ ⊥

The reason is that preservesPos′ requires the function argument to be positive, whereas
preservesPos says that the result is positive if and only if the argument was positive. ◦

We can add another contract constructor:

List :: Contract a→ Contract [a ]

The corresponding case of assert is as follows:

assert (List c) xs = map (assert c) xs

7 (8 points). Consider

allPos = List pos
allPos′ = Pred (all (>0))

Describe the differences between assert allPos and assert allPos′, and more generally
between using List versus using Pred to describe a predicate on lists. (Hint: Think
carefully and consider different situations before giving your answer. What about using
the allPos and allPos′ contracts as parts of other contracts? What about lists of functions?
What about infinite lists? What about strict and non-strict functions working on lists?)
[No more than 60 words.] •

Solution 7. The differences are due to laziness:

test1 c = length (assert c [−1 ])

Now:

test1 allPos ≡ 1
test1 allPos′ ≡ ⊥ (contract violation)

Another situation:

test2 c = take 10 (assert c [1 . . ])

Now:

test2 allPos ≡ [1, 2, 3, 4, 5, 6, 7, 8, 9, 10 ]
test2 allPos′ ≡ ⊥ (nontermination)

The contract allPos applies pos to the list elements lazily and can therefore miss errors,
whereas allPos′ forces evaluation and can thereby change the strictness behaviour of the
program.

Generally, Pred is more flexible than List because it can describe properties that are not
uniform over the list elements. However, List can be combined with function contracts,
whereas Pred cannot. ◦

8 (8 points). Discuss the advantages and disadvantages of using contracts and using
QuickCheck properties. What is similar, what are the differences? [No more than 60
words.] •

Solution 8. Some advantages of QuickCheck: automatic generation of test cases, can
be used to formulate algebraic properties that relate several functions. Disadvantages:
test data may not reflect actually used data, has to be run explicitly.

Some advantages of Contracts: actual program runs are checked, allows for design
by contract (for instance, blame assignment possible), can easily be switched off. Dis-
advantages: only actualy program runs are checked, difficult to express interaction
between several functions. ◦

9 (5 bonus points). Can contracts be translated into QuickCheck properties automati-
cally? If yes, try to define a function that does this. If not, discuss the difficulties. [No
more than 60 words.] •

Solution 9. In short: Translation for predicates is easy. They contain boolean properties
that are immediately testable. Translation for first-order functions is also possible if
the domain type is in the class Arbitrary. We can then map a function contract to a
property that generates arbitrary candidate values, and rejects those that do not fulfill
the precondition. Higher-order functions are not so easy to translate. ◦

Maps and folds (29 points total)

10 (8 points). For all f , g and z of suitable type, the equation

foldr f z ◦map g ≡ foldr (f ◦ g) z

holds. Prove this theorem using equational reasoning and induction on lists. •

Solution 10. Case [ ]:

(foldr f z ◦map g) [ ]
≡ { definition of (◦) }

foldr f z (map g [ ])
≡ { definition of map }

foldr f z [ ]
≡ { definition of foldr }

[ ]
≡ { definition of foldr }

(foldr (f ◦ g) z) [ ]

Case x : xs:

(foldr f z ◦map g) (x : xs)
≡ { definition of (◦) }

foldr f z (map g (x : xs))
≡ { definition of map }

foldr f z (g x : map g xs)
≡ { definition of foldr }

f (g x) (foldr f z (map g xs))
≡ { definition of (◦), twice }

(f ◦ g) x ((foldr f z ◦map g) xs)
≡ { induction hypothesis }

(f ◦ g) x ((foldr (f ◦ g) z) xs)
≡ { definition of foldr }

(foldr (f ◦ g) z) (x : xs)

◦

11 (6 points). Translate the following program into System F, i.e., make all type abstrac-
tions and type applications explicit, and annotate all value-level lambda abstractions
with their types.

mm :: (a→ b)→ [ [a ] ]→ [b ]
mm = λf xss→ head (map (map f ) xss)

(Hint: It is not necessary to translate head and map, but writing down their System F
types with explicit quantification will help you to know where to put type arguments.)

•

Solution 11. Type arguments are in brackets. Syntax wasn’t important, though, as long
as the abstractions and applications were properly indicated.

mm = λ〈a〉 〈b〉 (f :: a→ b) (xss :: [ [a ] ])→
head 〈[b ]〉 (map 〈[a ]〉 〈[b ]〉 (map 〈a〉 〈b〉 f ) xss)

◦
The following data type is known as a generalized rose tree:

data GRose f a = GFork a (f (GRose f a))

12 (3 points). What is the kind of GRose? •

Solution 12.

GRose :: (∗ → ∗)→ ∗ → ∗

◦
If we instantiate f to [ ], we get a rose tree, a tree that in every node can have arbitrarily

many subtrees. Leaves can be represented by choosing an empty list:

leaf :: a→ GRose [ ] a
leaf x = GFork x [ ]

13 (6 points). What if we instantiate f to Identity (where

newtype Identity a = Identity a

is the identity on the type level)? And what if we instantiate f to Maybe? What kind of
trees do we get, and what kind of familiar data structures are they similar to? [No more
than 40 words.] •

Solution 13. In the case of Identity we get trees where every node has exactly one child.
This is similar to streams (infinite lists). In the case of Maybe, every node can have one
child or no children. This is like a non-empty list. ◦

14 (6 points). Define an instance of class Functor for GRose, assuming that f is a Functor,
and defining a function fmap such that the passed function is applied to all the elements
of type a. •

Solution 14.

instance (Functor f )⇒ Functor (GRose f ) where
fmap f (GFork x xs) = GFork (f x) (fmap (fmap f ) xs)

◦

Simulating inheritance (23 points total)

Using open recursion and an explicit fixed-point operator similar to

fix f = f (fix f )

we can simulate some features commonly found in OO languages in Haskell. In many
OO languages, objects can refer their own methods using the identifier this, and to
methods from a base object using super.

We model this by abstracting from both this and super:

type Object a = a→ a→ a
data X = X {n :: Int, f :: Int→ Int}
x, y, z :: Object X
x super this = X {n = 0, f = λi→ i + n this}
y super this = super {n = 1}
z super this = super { f = f super ◦ f super}

We can extend one “object” by another using extendedBy:

extendedBy :: Object a→ Object a→ Object a
extendedBy o1 o2 super this = o2 (o1 super this) this

By extending an object o1 with an object o2, the object o1 becomes the super object for o2.
Once we have built an object from suitable components, we can close it to make it

suitable for use using a variant of fix:

fixObject o = o (error "super") (fixObject o)

We close the object o by instantiating it with an error super object and with itself as this.

15 (3 points). What is the (most general) type of fixObject? •

Solution 15. It really is

fixObject :: (a→ b→ b)→ b

but

fixObject :: (a→ a→ a)→ a

or equivalently

fixObject :: Object a→ a

are morally correct. ◦

16 (8 points). What are the values of the following expressions?

n (fixObject x)
f (fixObject x) 5
n (fixObject y)
f (fixObject y) 5
n (fixObject (x ‘extendedBy‘ y))
f (fixObject (x ‘extendedBy‘ y)) 5
f (fixObject (x ‘extendedBy‘ y ‘extendedBy‘ z)) 5
f (fixObject (x ‘extendedBy‘ y ‘extendedBy‘ z ‘extendedBy‘ z)) 5

•

Solution 16. In order: 0, ⊥ (but 1 is morally correct), ⊥, 1, 6, 7, 9. ◦

17 (4 points). Define an object

zero :: Object a

such that for all types t and objects x :: Object t, the equation x ‘extendedBy‘ zero ≡
zero ‘extendedBy‘ x ≡ x hold. [No proof required, just the definition.] •

Solution 17.

zero super this = super

Here is the proof for completeness:

x ‘extendedBy‘ zero
≡

(λsuper this→ zero (x super this) this)
≡

(λsuper this→ x super this)
≡

x

◦

A more interesting use for these functional objects is for adding effects to functional
programs in an aspect-oriented way.

In order to keep a function extensible, we write it as an object, and keep the result
value monadic:

fac :: Monad m⇒ Object (Int→ m Int)
fac super this n =

case n of
0 → return 1
n→ liftM (n∗) (this (n− 1))

Note that recursive calls have been replaced by calls to this. We can now write a separate
aspect that counts the number of recursive calls:

calls :: MonadState Int m⇒ Object (a→ m b)
calls super this n =

do
modify (+1)
super n

We can now run the factorial function in different ways:

runIdentity (fixObject fac 5) ≡ 120
runState (fixObject (fac ‘extendedBy‘ calls) 5) 0 ≡ (120, 6)

18 (8 points). Write an aspect trace that makes use of a writer monad to record whenever
a recursive call is entered and whenever it returns. Also give a type signature with the
most general type. Use a list of type

data Step a b = Enter a
| Return b

deriving Show

to record the log. As an example, the call

runWriter (fixObject (fac ‘extendedBy‘ trace) 3)

yields

(6, [Enter 3, Enter 2, Enter 1, Enter 0, Return 1, Return 1, Return 2, Return 6 ])

•

Solution 18.

trace :: MonadWriter [Step a b ] m⇒ Object (a→ m b)
trace super this a =

do
tell [Enter a ]
b← super a
tell [Return b ]
return b

In fact, trace has an even more general type, but the type above was sufficient. ◦

